THE GREENHOUSE EFFECT GAME BY MICROCOMPUTER

Nario Baba*

January 1984
PP-84-1

*Faculty of Engineering, Tokushima University,
Josanjima, Tokushima City, Japan

INTERNATIONAL INSTITUTE FOR APPLIED SYSTEMS ANALYSIS
A-2361 Laxenburg, Austria

ACKNOWLEDGEMENTS

The author would like to thank Prof. Y. Sawaragi for his kind advice. He would also like to express his gratitude to Mr. Kaji and Mr. Hashimoto for their assistance.

ABSTRACT

In 1981 Jesse Ausubel and his associates at IIASA developed a simple board game to illustrate some of the causes and consequences of human actions that lead to the accumulation of CO_2 in the atmosphere. In this paper is presented an adaptation of that game where the board and some of the accounting is implemented in N-Basic on a NEC 9801 microcomputer. The computer program is listed in the paper for those who might wish to adapt it for other machines.

The computerized version of the game was played on two occasions by students of the Faculty of Engineering, Tokushima University, Japan. Vague generalizations about the impact of the game are offered on the basis of this limited sample.

CONTENTS

THE GREENHOUSE EFFECT GAME BY MICROCOMPUTER

Norio Baba

1. INTRODUCTION

Human activities have strong effects upon global climatic changes. In particular, growing emissions of carbon dioxide into the atmosphere may cause the warming of the earth. If mankind pursues only the increase in GNP and does not care about the potential for environmental catastrophe, the ice of the antarctic could eventually be melted down. This would raise the level of the ocean, with really dreadful effects upon human beings. Recently, an IIASA research group made a very interesting educational board game called *Greenhouse Effect*, which is intended to raise public consciousness about the importance of the CO_2 issue.

In this paper, we present a microcomputer-based version of the CO_2 board type game. Since this game is in a dialogue mode and employs limited color graphics, it could be easier to learn and more interesting to play than the original version.

2. MICROCOMPUTER GAMING SYSTEM

The program is written in N-Basic, which is fitted to the NEC 9801 Microcomputer System. This program can be easily adapted to other microcomputer systems, for example the Apple II.

The game is quite similar to the original game. It can be played with 2, 3, or 4 players. Players take turns throwing a 6-sided die twice. The number is entered into the computer, and the screen indicates the status of the earth, Figure 1. The combination of the faces of the two turns, between 0 and 5, decides the growth in global GNP. There are fifty-nine states of the earth which consist of various kinds of spaces (impact



Figure 1. The microcomputer screen shows that the earth has advanced to stage 17, which offers players an opportunity to purchase adaptive measures that will protect against later costly impacts.

space, adaptation space, scientific uncertainty, etc.) (Since the rules and definitions of this game are almost the same as those of the original board game, please refer to the user's manual, Ausubel 1981.)

One can see lots of beautiful color graphics on the screen. When the earth moves towards the 59th state, sometimes the wave on the screen becomes very rough and high (Figure 2). When the game is ended by a CO_2-induced environmental catastrophe, the computer program illustrates a globe in which the Antarctic is melted down, and all the continents immerse into the sea (Figure 3).

3. THE PLAYING OF THE GAME BY JAPANESE STUDENTS

The first experiment with this game was carried out in the microcomputer room of Information Science and Systems Engineering Faculty of Engineering, Tokushima University, Japan, on the evening of February 23, 1983 and also on the morning of February 24, 1983.

Four students of Tokushima University and one of the authors participated. It was not a sample representative of the population at large, since all of the students were involved with research on game theory and gaming. They had read over the paper "Greenhouse Effect" with one of the authors and understood quite well the microcomputer program.

Fifteen game runs were carried out. Generally speaking, almost all of the players placed too much emphasis on increasing their GNP chips and too little resources into using prevention units to avoid the growth in CO_2. In the 2nd, 4th, 7th, 8th and 14th game, global catastrophies occurred. However, all of the players gradually recognized the real meaning of this game, and they used more prevention units in the latter half of the games. They also cooperated to manoeuver the earth into various spaces that afforded more control. These results indicated the educational effect of this game. The game runs were followed by a discussion with the gaming participants. The students agreed that this game has an educational effect for the CO_2 issue and is also very interesting. They also expressed keen interest in playing a more realistic game.

The second experiment was carried out in the same room on the morning of February 25, 1983. Four 2nd-year students and one 4th-year student participated. In this experiment, only 2 game runs were done. In the first run, catastrophe occurred, and all players lost all of their GNP chips. They paid no consideration to the state of the earth. Prevention units were only used at the 17th, 19th, and 20th turns. Players had a strong interest to get many GNP chips. The second game run followed a 5 minute break. In this game, all players were very cooperative, and they used many GNP chips to preserve the earth's climate.



Figure 2. As players near the end of the game, an illustration of a wave breaking over the shore warns of impending serious ecological damage.

Figure 3. There are two possible endings for the game. In one fossil fuels are exhausted before CO_2 mounts to catastrophic levels. In the second, illustrated here by the microcomputer, the Antarctic ice cap slowly melts and the continents subside partially beneath the sea.

REFERENCES

Ausubel, J.H. (1981) *The Greenhouse Effect: An Educational Board Game*. Instruction Booklet. Laxenburg, Austria: International Institute for Applied Systems Analysis.

APPENDIX: COMPUTER LISTING

```
5 CLS 3
6 CONSOLE ,,0,1
7 COLOR 4
10 PRINT "***GREEN HOUSE EFFECT GAME***"
15 PRINT
20 PRINT "The Green House Effect Game is an educational game intended to help pe
ople learn about carbon deoxide issue."
30 DIM G(5),W(5),A(5),C(5)
40 L5=0
45 K7=0
50 T7=0
60 Z1=0
65 COLOR 7
70 PRINT : INPUT "How many players ";N5
80 N2=12/N5
90 FOR K1=1 TO N5
100 G(K1)=N2
110 W(K1)=0
115 A(K1)=0
120 NEXT K1
130 PRINT : PRINT "Initial GNP chips and Prevention Units ."
140 FOR K2=1 TO N5
150 PRINT "Player";K2;"  G(";K2;")=";G(K2);"  W(";K2;")=";W(K2)
160 NEXT K2
170 I=1
180 PRINT
190 PRINT "The player :";I
200 IF T7<=23 THEN GOTO 240
202 PRINT "Do you want to use prevention unit ? (But,If TI=0, you can not use.)"
    : INPUT "   [yes-1, no-0]  U=" ; U
204 W(I)=W(I)-U
206 PRINT "Do you want to buy the prevention units for future turns ?"
   : INPUT"     [yes 1-1,2-2, no 0]  U2=" ; U2
210 PRINT
   :PRINT "Roll the die .  If 0 or 5 appears , our game is over ." :  GOSUB 7000
220 IF D2=0 OR D2=5 THEN GOTO 2615
235 GOTO 300
240 PRINT
250 PRINT "Do you want to use prevention unit ? (But, if TI=0,you can not use.)"
:INPUT "     [yes-1, no-0]  U=";U
260 W(I)=W(I)-U
270 PRINT : PRINT "Do you want to buy the prevention units for future turns ?"
  : INPUT"     [yes 1-1,2-2; no-0] U2=";U2
275 IF T7>23 THEN GOTO 300
280 PRINT :PRINT
290 PRINT"Roll the die . " : GOSUB 7000
```

```
295 PRINT :PRINT
300 INPUT "Values of U,U2 and D2 are OK ?  OK:1  NO:0  L6=";L6
302 IF L6<>1 THEN GOTO 250
308 Z1=Z1+D2-U
310 G(I)=G(I)+D2
320 PRINT : PRINT "The space of the earth=";Z1
  :PRINT  :PRINT "     Your total GNP chips=";G(I)
340 W(I)=W(I)+U2
350 G(I)=G(I)-U2
360 PRINT : PRINT "State of your GNP chips .":PRINT
362 FOR K2=1 TO N5
363 PRINT "Player :";K2;"   G(";K2;")=";G(K2)
364 NEXT K2
365 PRINT :PRINT
370 PRINT "If the total number of your GNP-chips is less than -10 , you must ret
ire .":PRINT "How about you ?  Can you continue to play ?"
    :INPUT "                   [yes-1,no-0]  K1=";K1
380 IF Z1<5 THEN GOTO 430
382 GOSUB 3000
384 CIRCLE(320,90),180,1
385 PAINT(320,175),2,1
386 FOR L=1 TO 3000
387 REM
388 NEXT L
390 PAINT(320,178),1,1 : PAINT(421,147),1,1  : PAINT(430,142),1,1
391 PAINT(432,107),1,1 : PAINT(360,103),1,1 : PAINT(330,101),1,1
392 PAINT(206,122),1,1 : PAINT(384,40),1,1
393 PAINT(368,60),1,1 : PAINT(375,56),1,1  : PAINT(382,51),1,1
394 FOR I=1 TO 5000
395 REM
396 NEXT I
397 CLS 3
398 PRINT :PRINT : COLOR 2
400   PRINT "*********************************************"
    : PRINT "*                                           *"
    : PRINT "*  Game is over .                           *"
405   PRINT "|  Environmental catastrophe has occurred . |"
    : PRINT "|  All players lose their GNP-chips .       |"
    : PRINT "|                                           |"
407   PRINT "*********************************************"
410 PRINT :PRINT : COLOR 7
420 GOTO  2710
430 PRINT :PRINT
440 IF Z1=0 THEN GOTO  2500
450 ON Z1 GOTO  2500,2500,2500,2500,1040
460 ON Z1-5 GOTO  1040,2500,1560,2500,780
470 ON Z1-10 GOTO  2500,600,1420,1560,1040
480 ON Z1-15 GOTO  1040,1420,780,600,600
490 ON Z1-20 GOTO  2500,2500,1560,2500,1560
500 ON Z1-25 GOTO  780,2500,600,1040,1040
510 ON Z1-30 GOTO  2500,780,600,2500,1040
520 ON Z1-35 GOTO  1210,1210,600,1420,1420
530 ON Z1-40 GOTO  780,600,600,1560,2500
540 ON Z1-45 GOTO 1560,2500,780,2500,1210
550 ON Z1-50 GOTO  1210,2500,1560,600,2500
560 ON Z1-55 GOTO 2500,1210,2500
570 '
580 '
590 '
600 REM ***IMPACT***
602 GOSUB 6000
603 WIDTH 40
604 LOCATE 14,9 : PRINT "IMPACT"
605 LOCATE 15,10 : PRINT "SPACE"
607 FOR LL1=1 TO 6000
608 REM
609 NEXT LL1 : CLS 3 :WIDTH 80
610   PRINT "    ***IMPACT***                COLOR-Z1"
    : PRINT "       BIG RED:1                   1-42,43"
    : PRINT "    MEDIUM RED:2                   2-19,20,54"
612   PRINT "     SMALL RED:3                   3-28"
    : PRINT "   SMALL GREEN:4                   4-38"
    : PRINT " MEDIUM GREEN:5                   5-12"
613   PRINT "     BIG GREEN:6                   6-33"
615 PRINT "The space of the earth : ";Z1
620 PRINT : IF Z1=38 OR Z1=12 OR Z1=33 THEN GOTO  660
621 IF Z1=42 OR Z1=43 THEN GOTO  710
622 INPUT "Do you have adaptation units ?   [yes-1 , no-0]   ";O2
623 IF O2<>0 THEN GOTO  665
```

```
640 INPUT "Is there any player who has adaptation unit and wants to specula
RED only )?  [yes-number of the player ; no-0 ]";M
645 PRINT
650 PRINT "Roll the die . (If the number is less than zero, you must  pay to
 speculator .)" : INPUT "How much ?   X,Y=";X,Y :GOSUB 7900
652 IF M=0 THEN GOTO 662
657 IF A(M)<>0 THEN GOTO 668
658 PRINT CHR$(7)
659 PRINT "You have no adaptation unit ." :GOTO 640
660 PRINT "Roll the die .": INPUT "How much ?  X,Y=";X,Y :GOSUB 7900
662 G(I)=G(I)+D3
663 PRINT : COLOR 3 : PRINT "Then your total GNP-chips .":PRINT "Player :";I;"
 GNP-chips=";G(I) : COLOR 7
664 GOTO 2500
665 PRINT "Do you use 1 adaptation unit ?":INPUT "If you use it , you will not l
ose GNP-chips .   [yes-1 , no-0]    ";O5
666 IF O5<>1 THEN GOTO 640
667 A(I)=A(I)-1:GOTO 2500
668 G(I)=G(I)+D3
670 IF D3>0 THEN GOTO 690
680 G(M)=G(M)-D3
685 A(M)=A(M)-1
690 PRINT : COLOR 3
  : PRINT "Then , total GNP-chips  ": PRINT "Player :";I;"   GNP-chips=";G(I);"
    Player :";M;"  GNP-chips=";G(M) : COLOR 7
700 GOTO 2500
710 FOR K=1 TO 100
720 BEEP 1
730 NEXT K: BEEP 0
740 GOTO 622
750 '
760 '
770 '
780 REM  **EXTERNAL FACTORS**
782 GOSUB 6000
783 WIDTH 40
784 LOCATE 13,9 :PRINT "EXTERNAL"
785 LOCATE 14,10 :PRINT "FACTOR"
787 FOR LL2=1 TO 6000
788 REM
789 NEXT LL2 : CLS 3 :WIDTH 80
790 PRINT "**EXTERNAL FACTORS**"
800 INPUT "Roll the die .  Write the result .   X,Y=";X,Y
805 GOSUB 7500
810 ON R1 GOTO 870,930,970
820 INPUT "All players must roll the die .   Who is highest ?   J=";J
830 COLOR 5 : PRINT "You can get 1 GNP-chip ."  : COLOR 7
840 G(J)=G(J)+1
850 PRINT "Your total GNP-chips :  G(";J;")=";G(J)
860 GOTO 2500
870 COLOR 6 : PRINT "All players lose 1 GNP-chip ." : COLOR 7
880 FOR K=1 TO N5
890 G(K)=G(K)-1
900 NEXT K
920 GOTO 2500
930 COLOR 6 : PRINT "*Advance earth 3 steps ." : COLOR 7
940 Z1=Z1+3
950 PRINT "The location of the earth is : Z1=";Z1
960 GOTO 2500
970 INPUT "All players must roll the die .   Who is lowest ?   J=";J
980 COLOR 6 : PRINT "You lose 2 GNP-chips ." : COLOR 7
990 G(J)=G(J)-2
1000 PRINT "Your total GNP-chips : G(";J;")=";G(J)
1010 GOTO 2500
1020 '
1022 '
1025 '
1030 '
1040 REM   ***ADAPTATION SPACE***
1042 GOSUB 6000
1043 WIDTH 40
1044 LOCATE 12,9 :PRINT "ADAPTATION"
1045 LOCATE 15,10 :PRINT "SPACE"
1047 FOR LL3=1 TO 6000
1048 REM
1049 NEXT LL3 : CLS 3:WIDTH 80
1050 PRINT "**Adaptation Space**"
1060 PRINT "You can purchase 1 blue adaptation unit for 1 GNP chip or 2 units fo
r 3 GNP chips."
    :  PRINT "If you land on an impact space, this chip is very helpful. "
```

```
1065 PRINT "You can also speculate against another player ."
1070 INPUT "How many chips do you want to buy? (0,1,2)  A.U.= " ; A1
     :  PRINT
1072 IF A1=0 THEN GOTO 2500
1080 IF A1=1 THEN GOTO 1130
1090 IF A1=2 THEN GOTO 1110
1100 GOTO 1070
1110 G(I)=G(I)-3
1120 GOTO 1140
1130 G(I)=G(I)-1
1140 PRINT "Your total GNP chips :" ; G(I)
1145 A(I)=A(I)+A1
1150 GOTO 2500
1160 '
1170 '
1180 '
1200 '
1210 REM  ***COMPENSATION***
1212 GOSUB 6000
1213 WIDTH 40
1214 LOCATE 12,9 :PRINT "COMPENSATION"
1215 LOCATE 15,10 :PRINT "SPACE"
1217 FOR LL4=1 TO 6000
1218 REM
1219 NEXT LL4 : CLS 3:WIDTH 80
1220   PRINT "**Compensation Space**   1chip = back 1space"
     : PRINT "                         4chip = back 2spaces"
     : PRINT "                         9chip = back 3spaces"
1230 PRINT "How much do you want to pay ?  You must pay first."
1235   PRINT "Player" ; I ; "=" ;
     : INPUT C(I)
1240 PRINT
1250 : PRINT "Is there any player who is willing to pay ? "
     : PRINT "Write your payments in the regular order. "
1260 FOR K2 = 1 TO N5
1261 IF I <> K2 THEN GOTO 1263
1262   PRINT "Player" ; I;"=" ; C(I)
     : GOTO 1266
1263   PRINT "Player" ; K2 ; "=" ;
     : INPUT C(K2)
1266 NEXT K2
1268 C7 = 0
1269 FOR K4 = 1 TO N5
1270 C7 = C7+C(K4)
1271 NEXT K4
1272 IF C7 = 0 THEN GOTO 1279
1273 IF C7 = 1 THEN GOTO 1280
1274 IF C7 = 4 THEN GOTO 1282
1276 IF C7 = 9 THEN GOTO 1284
1278 GOTO 1260
1279   B1 = 0
     : GOTO 1290
1280   B1 = 1
     : GOTO 1290
1282   B1 = 2
     : GOTO 1290
1284   B1 = 3
     : GOTO 1290
1290 PRINT "Then,the earth backs" ; B1 ; "spaces."
1300 Z1 = Z1-B1
1310 PRINT "The space of the earth is" ; Z1
1320 FOR K2 = 1 TO N5
1330 G(K2) = G(K2)-C(K2)
1340 NEXT K2
1360 GOTO 2500
1370 '
1380 '
1390 '
1400 '
1410 '
1420 REM  ***FORESEEABLE FACTORS***
1422 GOSUB 6000
1423 WIDTH 40
1424 LOCATE 12,9 :PRINT "FORESEEABLE"
1425 LOCATE 15,10 :PRINT "FACTOR"
1427 FOR LL5=1 TO 6000
1428 REM
1429 NEXT LL5 : CLS 3:WIDTH 80
```

```
1430   PRINT " **Foreseeable Factors**"
     : PRINT "If you don't pay 1GNPchip,you must pay lots of GNPchips"
     : PRINT "at #19,20,54,42,43,M.B,(Impact Spaces) 46(Severe S.U.)."
1432 PRINT "But,if you pay now,you need not pay more than 1 GNP chip ."
1435 IF L5=0 THEN GOTO  1440
1437 PRINT "You need not pay 1 GNP chip because you have already paid ." :PRINT
 :PRINT : GOTO 2500
1440 PRINT "Do you pay 1 GNP-chip ?   Write 1 for yes and write 0 for no ."
1445 INPUT "I5(Who)=";I5  :INPUT "F1(1 or 0)=";F1
1450 A5=G(I5)-F1
1460    PRINT "How about other players?"
      : PRINT "If you pay 3GNP chips (total) now."
      : PRINT "you need not pay at #46 and the earth advances only 3 steps at #2
3."
1470 INPUT "All other players agree to pay?   [Yes-1,No-0] : L1=?" ; L1
1480 IF L1=0 THEN GOTO 1540
1485 L5=L5+L1
1490 FOR K=1 TO N5
1500 G(K)=G(K)-L1
1510 NEXT K
1520 G(I5) = A5-1
1530 GOTO 2500
1540 INPUT"No cooperation.   Who disagrees to cooperate?  k7=?  (All:0)  " ;K7
1545 G(I5)=A5
1550 GOTO 2500
1560 REM **SCIENTIFIC UNCERTAINTY**
1562 GOSUB 6000
1563 WIDTH 40
1564 LOCATE 12,9 :PRINT "SCIENTIFIC"
1565 LOCATE 12,10 :PRINT "UNCERTAINTY"
1567 FOR LL6=1 TO 6000
1568 REM
1569 NEXT LL6 : CLS 3:WIDTH 80
1570 PRINT " **Scientific Uncertainty**        COLOR-Z1 "
1572 PRINT "   Serious. S.U. :1                    1-23":
    : PRINT "   Severe.  S.U. :2                    2-46":
    : PRINT "      Red.  S.U. :3                     3-14,53"
1575   PRINT "    Green.  S.U. :4                     4-8"
     : PRINT "     Good.  S.U. :5                     5-44"
     : PRINT "      Bad.  S.U. :6                     6-25"
1580 IF Z1 <> 23 THEN GOTO 1630
1585 PRINT "*Serious. S.U.*"
1590 FOR K=1 TO 200
1600 BEEP 1
1610 NEXT K
1620 BEEP 0
1630 PRINT
1640 PRINT "Z1=" ; Z1
1660 IF Z1 <> 23 THEN GOTO 1720
1670 IF L5 = 0 THEN GOTO 1700
1680 Z1=26 : COLOR 2 : PRINT "The earth advances 3 steps."
1685 L5=L5-1
1690 GOTO 2500
1700 Z1 = 31 : COLOR 2 : PRINT "The earth advances 8 steps."
1710 GOTO 2500
1720 IF Z1 <> 46 THEN GOTO 2150
1725 PRINT "*Severe S.U.*"
1730   PRINT "Now we are in the space of Severe scientific uncertainty."
     : PRINT "Player" ; I : "rolled the die."
     : COLOR 3
1740 D7 = G(I)
1750 D8 = G(I5)
1760 IF K7 = 0 THEN  GOTO 1780
1770 D9 = G(K7)
1780 IF F1 <> 1 THEN GOTO 1970
1790 IF L5 <> 0 THEN GOTO 1940
1800 IF K7 <> 0 THEN GOTO 1870
1810 FOR K = 1 TO N5
1820 G(K) = G(K)-3
1830 NEXT K
1840 G(I) = D7-4
1850 G(I5) = D8-1
1852 IF I <> 15 THEN GOTO 1855
1853   PRINT "Player" ; I ; "lost 1 GNP chip."
     : GOTO 1856
1855 PRINT "Player" ; I : "lost 4 GNP chips and player";I5; "lost 1 GNP chip."
1856 PRINT "All other player lost 3 GNP-chips ."
1860 GOTO 2500
1870 FOR K=1 TO N5
1880 G(K)=G(K)-2
```

```
1890 NEXT K
1900 G(I)=D7-4
1910 G(I5)=D8-1
1920 G(K7)=D9-4
1921 IF I<>I5 THEN GOTO 1923
1922 PRINT "Player";I;"lost 1 GNP-chip and player";K7;"lost 4 GNP-chips ."
     :GOTO 1926
1923 IF I<>K7 THEN GOTO 1925
1924 PRINT "Player";I;"lost 4 GNP-chips and player";I5;"lost 1 GNP-chip ."
     :GOTO 1926
1925 PRINT "Player";I;"and player";K7;"lost 4 GNP-chips and player";I5;"lost 1 G
NP-chip ."
1926 PRINT "All other player lost 2 GNP-chips ."
1930 GOTO 2500
1940 IF I5<>1 THEN GOTO 1965
1950 G(I5)=D8-1
1955 PRINT "Player";I;"lost 1 GNP-chip ."
1960 GOTO 2500
1965 PRINT "All players have no influence ."  :GOTO 2500
1970 IF L5<>1 THEN GOTO 2030
1980 IF I5<>1 THEN GOTO 2010
1990 G(I5)=D8-4
1995 PRINT "Player";I;"lost 4 GNP-chips ."
2000 GOTO 2500
2005 PRINT "Player";I5;"lost 3 GNP-chips ."
2010 G(I5)=D8-3
2015 PRINT "Player";I5;"lost 3 GNP-chips ."
2020 GOTO 2500
2030 IF K7=0 THEN GOTO 2100
2040 FOR K=1 TO N5
2050 G(K)=G(K)-2
2060 NEXT K
2070 G(I)=D7-4
2080 G(K7)=D9-4
2082 IF I<>K7 THEN GOTO 2084
2083 PRINT "Player";I;"lost 4 GNP-chips ."   :GOTO 2085
2084 PRINT "Player";I;"and player";K7;"lost 4 GNP-chips ."
2085 PRINT "All other player lost 2 GNP-chips ."
2090 GOTO 2500
2100 FOR K=1 TO N5
2110 G(K)=G(K)-3
2120 NEXT K
2130 G(I)=D7-4
2132 PRINT "Player";I;"lost 4 GNP-chips ."
2135 PRINT "All other player lost 3 GNP-chips ."
2140 GOTO 2500
2150 IF Z1<>8 THEN GOTO 2260
2160 '
2170 '
2180 '
2190 '
2200 '
2210 PRINT "Green Scientific Uncertainty ."
     :PRINT "Roll the Medium Green die ."
2220 INPUT "Number: X,Y=";X,Y :GOSUB 7900
2230 Z1=Z1-D3
2240 COLOR 4 : PRINT "Back the earth";D3;"steps ." : COLOR 7
2250 GOTO 2500
2260 IF Z1<>44 THEN GOTO 2310
2270 PRINT "Good Scientific Uncertainty"
2280 G(I)=G(I)+1
2290 COLOR 1 : PRINT "You added 1 GNP-chip .   Your total :";G(I) : COLOR 7
2300 GOTO 2500
2310 IF Z1<>25 THEN GOTO 2410
2320 '
2330 '
2340 '
2350 '
2360 '
2370 PRINT "*Bad S.U.*"
2380 G(I)=G(I)-1
2390 COLOR 3 : PRINT "You lost 1GNP chip. Your total is" ; G(I) : COLOR 7
2400 GOTO 2500
2410 IF Z1 = 14 THEN GOTO 2440
2420 IF Z1=53 THEN GOTO 2440
2430 GOTO 2500
2440 PRINT "RED S.U.  Roll the Medium red die."
2450 INPUT "NUMBER :X,Y=" ;X,Y :GOSUB 7900
2460 Z1=Z1+ABS(D3)
2470 COLOR 2 : PRINT "The earth advanced"  ; N1 ;"steps." : COLOR 7
```

```
2500 REM
2501 FOR LL9=1 TO 8000
2502 NEXT LL9
2507 COLOR 7
2508 PRINT
2510   PRINT "Total Turns=" ; T7
     : PRINT "The space of the earth=" ; Z1
2520 PRINT "*Results*"
2530 FOR K2=1 TO N5
2540 PRINT "Player"  ; K2 ; "   GNP=" ; G(K2) ; "   P.U.=" ; W(K2) ; "   A.U.="
; A(K2)
2550 NEXT K2
2551 FOR LL9=1 TO 8000
2552 REM
2553 NEXT LL9
2556 IF Z1=34 OR Z1=22 THEN GOTO 2558
2557 IF Z1<45 THEN GOTO 2550
2558 GOSUB 5100
2560 T7=T7+1
2570 I=I+1
2580 IF I <= N5 THEN GOTO 180
2590 I=1
2600 GOTO 180
2610 RETURN
2615 GOSUB 3000
2616 FOR I=1 TO 4000
2617 REM
2618 NEXT I
2622 CLS 3
2623 WIDTH 40 :LINE(100,50)-(500,130),4,BF
2624 LOCATE 11,7 :PRINT "Technological"
2625 LOCATE 12,9 :PRINT "Breakthrough"
2626 FOR K=1 TO 5000
2627 REM
2628 NEXT K :CLS 3
2629 WIDTH 80
2640 PRINT "Your GNP chips"
2642 FOR K2=1 TO N5
2645 PRINT "   GNP(" ; K2 ; ")=" ; G(K2) ;
2648 NEXT K2
2650 GOTO 2710
2660 '
2670 '
2680 '
2690 '
2700 '
2710   PRINT
     : PRINT "Did you enjoy this game?   [yes-1,no-0]"
2712 FOR K2=1 TO N5
2713   PRINT "Player" ; K2 ; "=" ;
     : INPUT P1
2715 NEXT K2
2716   PRINT
     : PRINT
2720 PRINT "Do you want to improve this game?    [yes-1,no-0]"
2722 FOR K3=1 TO N5
2723   PRINT "Player" ; K3 ; "=" ;
     : INPUT P2
2725 NEXT K3
2726   PRINT
     : PRINT
2730 PRINT "How do you think about this game?  Please give us your comments(or a
dvices)."
2735 PRINT "If you have,push1.  If you have not, push 0."
2740   INPUT"P=" ; P
2744 PRINT
2750 PRINT "Thank you for your kind assistances."
2770 END
3000 CLS 3
3010 CIRCLE(320,90),180,7
3015 RESTORE 3100
3020 A1=154
3030 A2=60
3040 READ B1,B2
3050 IF B1=0 THEN GOTO 3160
3060 LINE(A1,A2)-(B1,B2),7
3070 A1=B1
3080 A2=B2
3090 GOTO 3040
```

```
3100 DATA 154,65,150,70,148,78,150,85,165,85,175,87,178,90,180,100,190,110,200,1
20,205,123,214,120,220,112,224,108,228,100,240,90,248,86,240,84,230,83,222,70
3110 DATA 240,82,264,75,248,66,270,73,280,77,292,87,306,68,314,78,330,90,329,87,
322,80,324,76,333,82,340,77,338,69,330,65
3120 DATA 340,65,350,60,352,55,350,50,354,48,348,47,342,47,342,45,346,43,358,44,
360,51,370,48,362,40,368,30,360,27,358,23,362,18
3130 DATA 372,15,380,11,384,13,388,20,392,22,394,18,393,14,389,12,384,9,378,8,34
8,10,324,13,312,11,286,15,274,11,277,8,274,8,274,7,243,11,230,24,226,25,224,28
3140 DATA 222,30,226,33,233,31,240,34,244,30,242,26,250,20,258,18,259,21,252,25,
256,27,263,26,262,29,251,31,240,37,234,35,230,36,225,37,236,37
3150 DATA 223,39,198,38,197,39,192,40,193,42,190,44,172,41,171,52,178,52,200,46,
208,46,210,55,212,55,212,46,228,52,232,47,242,45,248,47,254,46,260,51,256,55,230
,53,230,57,220,60,202,58,190,55,166,55,154,60,0,0
3160 RESTORE 3250
3170 A1=377
3180 A2=124
3190 READ B1,B2
3200 IF B1=0 THEN GOTO 3260
3210 LINE(A1,A2)-(B1,B2),7
3220 A1=B1
3230 A2=B2
3240 GOTO 3190
3250 DATA 377,124,378,128,380,134,390,132,400,131,410,135,414,139,420,142,430,14
3,434,142,440,137,443,131,440,122,422,110,420,116,416,117,410,115,412,110,400,11
0,386,117,374,118,377,124,0,0
3260 LINE(420,148)-(428,145),7
3270 LINE(428,145)-(420,144),7
3280 LINE(420,144)-(420,148),7
3290 RESTORE 3380
3300 A1=432
3310 A2=109
3320 READ B1,B2
3330 IF B1=0 THEN GOTO 3390
3340 LINE(A1,A2)-(B1,B2),7
3350 A1=B1
3360 A2=B2
3370 GOTO 3320
3380 DATA 437,106,430,100,420,100,410,95,400,95,388,98,432,109,0,0
3390 RESTORE 3480
3400 A1=372
3410 A2=97
3420 READ B1,B2
3430 IF B1=0 THEN GOTO 3490
3440 LINE(A1,A2)-(B1,B2),7
3450 A1=B1
3460 A2=B2
3470 GOTO 3420
3480 DATA 371,91,352,90,350,94,342,96,343,100,352,104,360,105,366,99,372,97,0,0
3490 RESTORE 3580
3500 A1=340
3510 A2=106
3520 READ B1,B2
3530 IF B1=0 THEN GOTO 3590
3540 LINE(A1,A2)-(B1,B2),7
3550 A1=B1
3560 A2=B2
3570 GOTO 3520
3580 DATA 343,105,337,104,332,99,330,95,314,90,328,95,310,90,316,94,316,96,340,1
06,340,106,0,0
3590 RESTORE 3680
3600 A1=268
3610 A2=176
3620 READ B1,B2
3630 IF B1=0 THEN GOTO 3690
3640 LINE(A1,A2)-(B1,B2),7
3650 A1=B1
3660 A2=B2
3670 GOTO 3620
3680 DATA 264,172,268,169,277,167,278,165,272,164,271,165,268,159,284,164,286,16
8,292,170,293,173,300,173,302,170,306,169,310,166,316,166,320,170,333,168,338,17
1,354,169,366,170,378,171,382,175,0,0
3690 LINE(383,37)-(378,41),7
3700 LINE(378,41)-(380,43),7
3710 LINE(380,43)-(382,42),7
3720 LINE(382,42)-(384,44),7
3730 LINE(384,44)-(390,40),7
3740 LINE(390,40)-(383,37),7
3750 RESTORE 3840
```

```
3760 A1=380
3770 A2=45
3780 READ B1,B2
3790 IF B1=0 THEN GOTO 3850
3800 LINE(A1,A2)-(B1,B2),7
3810 A1=B1
3820 A2=B2
3830 GOTO 3780
3840 DATA 379,50,366,53,365,55,378,54,379,52,382,55,386,53,384,44,380,45,0,0
3850 LINE(364,57)-(365,61),7
3860 LINE(365,61)-(370,62),7
3870 LINE(370,62)-(371,60),7
3880 LINE(371,60)-(368,56),7
3890 LINE(368,56)-(364,57),7
3900 LINE(380,58)-(380,55),7
3910 LINE(380,55)-(372,55),7
3920 LINE(372,55)-(375,59),7
3930 LINE(375,59)-(377,57),7
3940 LINE(377,57)-(380,58),7
3950 PAINT(200,70),6,7
3960 PAINT(420,130),6,7
3970 PAINT(422,146),6,7
3980 PAINT(375,57),6,7
3990 PAINT(420,103),6,7
4000 PAINT(360,100),6,7
4010 PAINT(320,95),6,7
4020 PAINT(384,40),6,7
4030 PAINT(381,53),6,7
4040 PAINT(368,59),6,7
4050 PAINT(320,175),7,7
4060 PAINT(240,160),1,7
5000 RETURN
5010 '
5020 '
5100 REM **n-a-m-i**
5110 CLS 3
5120 LINE(10,10)-(630,190),7,B
5130 X1=10
5140 Y1=70
5150 RESTORE 5240
5160 FOR F1=1 TO 62
5170 READ S1
5180 X2=X1+10
5190 Y2=Y1+S1
5200 LINE(X1,Y1)-(X2,Y2),7
5210 X1=X2
5220 Y1=Y2
5230 NEXT F1
5240 DATA 1,1,2,2,1,0,1,2,4,5
5250 DATA 3,1,0,0,1,0,-1,0,4,5
5260 DATA 4,4,3,2,2,3,1,0,1,1
5270 DATA 1,1,2,1,0,0,1,1,2,0
5280 DATA 4,3,1,1,3,2,0,1,2,1
5290 DATA 3,1,0,2,2,1,1,0,1,1,2,1
5300 X3=10
5310 Y3=100
5320 RESTORE 5410
5330 FOR F2=1 TO 20
5340 READ S2
5350 X4=X3+10
5360 Y4=Y3+S2
5370 LINE(X3,Y3)-(X4,Y4),7
5380 X3=X4
5390 Y3=Y4
5400 NEXT F2
5410 DATA 1,0,0,-1,-1,0,1,0,1,-1
5420 DATA 2,-1,0,1,-1,0,-1,2,1,-1
5430 PAINT(20,110),6,7
5440 PAINT(20,80),4,7
5450 NK=62-Z1
5460 X5=10
5465 Y5=0
5470 RESTORE 5240
5480 FOR F3=1 TO NK
5490 READ S3
5500 X5=X5+10
5510 Y5=Y5+S3
5520 NEXT F3
5530 X5=X5-5
5540 Y5=Y5+70
```

```
5550 LINE(X5,Y5)-(630,Y5),7
5560 PAINT(620,160),1,7
5580 N2=7
5590 FOR F4=1 TO 40
5610 LINE(630,Y5-1)-(X5,Y5-1),N2,,&HA146
5620 LINE(630,Y5-2)-(X5-1,Y5-2),N2,,&H4A11
5630 FOR F5=1 TO 200
5640 REM
5650 NEXT F5
5660 IF N2=7 THEN 5680
5670 N2=7 :GOTO 5690
5680 N2=0
5690 NEXT F4
5700 CLS 3
5710 RETURN
5800 '
5900 '
6000 REM **BOARD**
6010 CLS 3
6020 LINE(30,10)-(620,50),7,B
6030 LINE(550,10)-(620,180),7,B
6040 LINE(50,140)-(620,180),7,B
6050 FOR FUM1=50 TO 550 STEP 20
6060 LINE(FUM1,10)-(FUM1,50),7
6070 LINE(FUM1,140)-(FUM1,180),7
6080 NEXT FUM1
6090 FOR FUM2=65 TO 140 STEP 15
6100 LINE(550,FUM2)-(620,FUM2),7
6110 NEXT FUM2
6120 IF Z1>26 THEN GOTO 6220
6130 RESTORE 6420
6140 SK=Z1*20+40
6150 FOR KA=0 TO Z1
6160 READ NON1
6170 NEXT KA
6180 PAINT(SK,20),NON1,7
6190 CIRCLE(SK,35),7,0
6200 PAINT(SK,35),0,0 :BEEP
6210 GOTO 6470
6220 IF Z1>32 THEN GOTO 6330
6230 SK=(Z1-27)*15+57
6240 RESTORE 6440
6250 ZZ1=Z1-26
6260 FOR KJ=0 TO ZZ1
6270 READ NON1
6280 NEXT KJ
6290 PAINT(600,SK),NON1,7
6300 CIRCLE(570,SK),7,0
6310 PAINT(570,SK),0,0 :BEEP
6320 GOTO 6470
6330 SK=580-(Z1-33)*20
6340 RESTORE 6450
6350 ZZ3=Z1-32
6360 FOR KS=0 TO ZZ3
6370 READ NON1
6380 NEXT KS
6390 PAINT(SK,150),NON1,7
6400 CIRCLE(SK,150),7,0
6410 PAINT(SK,150),0,0 :BEEP
6420 DATA 7,7,7,7,7,1,1,7,3,7,6,7,2,5
6430 DATA 3,1,1,5,6,2,2,7,7,3,7,3,6
6440 DATA 0,7,2,1,1,7,6
6450 DATA 0,2,7,1,4,4,2,5,5,6,2,2,3,7,3,7
6460 DATA 6,7,4,4,7,3,2,7,7,4,7,7
6470 LINE(160,80)-(400,120),NON1,B
6480 RETURN
6490 '
6500 '
7000 INPUT"Number of dice X,Y=" ; X,Y
7010 DIM S(6,6)
7020 S(1,1)=0 : S(1,2)=1 : S(1,3)=2
7030 S(1,4)=3 : S(1,5)=4 : S(1,6)=5
7040 S(2,1)=5 : S(2,2)=4 : S(2,3)=3
7050 S(2,4)=2 : S(2,5)=1 : S(2,6)=0
7060 S(3,1)=1 : S(3,2)=2 : S(3,3)=3
7070 S(3,4)=4 : S(3,5)=3 : S(3,6)=2
7080 S(4,1)=3 : S(4,2)=0 : S(4,3)=2
7090 S(4,4)=5 : S(4,5)=1 : S(4,6)=4
7100 S(5,1)=4 : S(5,2)=3 : S(5,3)=2
```

```
7110 S(5,4)=1 : S(5,5)=2 : S(5,6)=3
7120 S(6,1)=2 : S(6,2)=2 : S(6,3)=4
7130 S(6,4)=1 : S(6,5)=3 : S(6,6)=3
7140 D2=S(X,Y)
7142 PRINT "The number = " ; D2
7145 ERASE S
7150 RETURN
7500 REM ***EXTERNAL FACTOR***
7510 DIM S(6,6)
7520 RESTORE 7580
7530 FOR I1=1 TO 6
7540 FOR JJ=1 TO 6
7550 READ S(I1,JJ)
7560 NEXT JJ
7570 NEXT I1
7580 DATA 1,2,3,4,1,2
7590 DATA 3,4,4,3,2,1
7600 DATA 2,1,4,3,4,3
7610 DATA 1,2,2,1,3,4
7620 DATA 4,3,2,1,4,3
7630 DATA 4,3,1,2,1,2
7635 R1=S(X,Y)
7640 PRINT "   R1=";R1
7645 ERASE S
7650 RETURN
7700 '
7800 '
7900 DIM SS(6,6)
7910 IF Z1=42 OR Z1=43 THEN GOTO 8020
7920 IF Z1=19 OR Z1=20 OR Z1=54 THEN GOTO 8030
7930 IF Z1=28 THEN GOTO 8040
7940 IF Z1=38 THEN GOTO 8050
7950 IF Z1=12 THEN GOTO 8060
7960 IF Z1=33 THEN GOTO 8070
7970 IF Z1=8 THEN GOTO 8060
7980 IF Z1=14 OR Z1=53 THEN GOTO 8030
8020 RESTORE 8150 :GOTO 8080
8030 RESTORE 8200 :GOTO 8080
8040 RESTORE 8250 :GOTO 8080
8050 RESTORE 8300 :GOTO 8080
8060 RESTORE 8350 :GOTO 8080
8070 RESTORE 8400
8080 FOR III=1 TO 6
8090 FOR JJ1=1 TO 6
8100 READ SS(III,JJ1)
8110 NEXT JJ1
8120 NEXT III
8130 D3=SS(X,Y)
8150 DATA -4,-3,-2,-1,0,1,1,0,-1,-2,-3,-4,-3,-2,-1,0,-1,-2
8160 DATA -1,-4,-2,1,-3,0,0,-1,-2,-3,-2,-1,-2,-2,0,-3,-1,-1
8200 DATA -2,-2,-1,-1,0,0,0,0,-1,-1,-2,-2,-2,-1,-1,0,-1,-1
8210 DATA -1,-2,-1,0,-2,0,0,-1,-1,-2,-1,-1,-1,-1,0,-2,-1,-1
8250 DATA -2,-1,-1,0,0,1,1,0,0,-1,-1,-2,-1,-1,0,0,0,-1
8260 DATA 0,-2,-1,1,-1,0,0,0,-1,-1,-1,0,-1,-1,0,-1,0,0
8300 DATA -1,0,0,1,1,2,2,1,1,0,0,-1,0,0,1,1,1,0
8310 DATA 1,-1,0,2,0,1,1,1,0,0,0,1,0,0,1,0,1,1
8350 DATA 0,0,1,1,2,2,2,2,1,1,0,0,0,1,1,2,1,1
8360 DATA 1,0,1,2,0,2,2,1,1,0,1,1,1,1,2,0,1,1
8400 DATA -1,0,1,2,3,4,4,3,2,1,0,-1,0,1,2,3,2,1
8410 DATA 2,-1,1,4,0,3,3,2,1,0,1,2,1,1,3,0,2,2
8413 PRINT "D3=";D3
8415 ERASE SS
8420 RETURN
```